
Download or browse the vbSerialization.exe in the MSDN Online Code Center.

When building applications using objects, we are often faced with the requirement to treat all the various data within an object as a single unit. This comes into play, for instance, when you want to pass an object across the network—since you don't want to send each individual bit of object data one at a time across the network, but rather, all at once. 

It is also very useful if you want to implement an 'undo' function in our object; that is, the ability to reset all the object's data back to some stored set of values. Sure, we can store the value of each individual variable somewhere, but it would be a lot simpler to store a single value that represents the entire state of the object.

You might also use this idea to implement a cloning function—a function that makes an exact copy of your object. Again, we could copy each data element field by field to the new object, but it is much more efficient to copy all the data into the new object as a single unit.

Serialization is a key concept in allowing you to treat all of an object's data as a single unit. Serialization is the process of converting all the various elements of data within your object into a single element, known as a byte stream. You can reverse this process through deserialization—converting the byte stream back into individual data elements within the object.

Deep and Shallow Serialization

The .NET Framework supports two general types of serialization: shallow and deep. Shallow serialization is the process of converting the read-write property values of an object into a byte stream, and is the technique used by the XmlSerializer and Web Services. This is called shallow serialization, because it doesn't serialize the object's underlying data but only the data available through public read-write property methods.

Figure 1. Shallow serialization—the process of copying property values to a byte stream
Deep serialization is the process of converting the actual values stored in an object's variables into a byte stream. It is the technique used by the BinaryFormatter and SoapFormatter objects, and by .NET Remoting. It is also used in a limited form by the LosFormatter to generate the state data stored in Web Forms pages. 

Deep serialization is more thorough, since it will copy values that are stored in private variables. It provides a much more complete copy of the object's data than shallow serialization. This is the type of serialization I'll focus on in this column.

Figure 2. Deep Serialization—the process of copying object data to a byte stream
Additionally, deep serialization will serialize an entire object graph. In other words, if your object holds a reference to another object, or to a collection of other objects, all those objects will be included in the serialization process as well. This is very powerful, since many applications have object hierarchies—invoices with line items, orders with detail, customers with addresses, and so forth.

Figure 3. Serializing an object graph—many objects combined into one byte stream
Obviously, there are times when you might not want to serialize the entire graph, so it is possible to prevent specific variables—including object references—within your object from being serialized as part of deep serialization. We'll discuss this later in the column.

Simple Serialization

Objects typically maintain state in instance variables. For instance, the following class provides a simple representation of a home:

Public Class Home

  Private mstrAddress As String

  Private mintSize As Integer

  Private mdtBuilt As Date

  Public Sub New(ByVal Address As String, ByVal Size As Integer, _

      ByVal Built As Date)

    mstrAddress = Address

    mintSize = Size

    mdtBuilt = Built

  End Sub

  Public ReadOnly Property Address() As String

    Get

      Return mstrAddress

    End Get

  End Property

  Public ReadOnly Property Size() As Integer

    Get

      Return mintSize

    End Get

  End Property

  Public ReadOnly Property Age() As Integer

    Get

      Return DateDiff(DateInterval.Year, mdtBuilt, Now)

    End Get

  End Property

End Class

This class implements three properties that can be used by any code using an instance of this class. The data used by the class is stored in the instance variables declared at the top of the class. It is this data that must be moved across the network in order to pass the object by value.

Notice that the variable storing the build date is not directly exposed as a property. Instead, there is an Age property exposed that is derived from the underlying date. This is a common occurrence in object design, and illustrates a scenario where shallow serialization will produce a very different result from deep serialization.

In fact, since all the properties are read-only, shallow serialization is entirely useless for objects based on this class, since it only deals with read-write property methods. Deep serialization, however, is fully capable of converting this object's data into a byte stream.

We can tell .NET to enable serialization and deserialization for our object's data by using the <Serializable()> attribute. This attribute is applied to a class, and it tells .NET that we want any objects based on that class to be available for serialization. For instance, we can apply the attribute to our Home class as follows:

<Serializable()> _

Public Class Home

By default, objects are not serializable—they are unavailable for deep serialization. Objects are always available for shallow serialization, since that technique merely scans the object's public interface for read-write property methods.

To enable deep serialization, we need to apply this <Serializable()> attribute. An advanced alternative to using this attribute is to implement the ISerializable interface, which I'll discuss later in the column.

Once a class is marked as serializable, we can use the capability built into .NET to convert our objects into a byte stream. Within .NET this byte stream can be placed in any object that derives from System.IO.Stream—the base stream data type in the system class library. This includes many useful types of stream, including memory streams, TCP/IP network streams, files on disk, and others.

The serialization and deserialization are handled by a special .NET object called a BinaryFormatter, which is found in the System.Runtime.Serialization.Formatters.Binary namespace. The BinaryFormatter object provides Serialize and Deserialize methods that allow us to easily serialize our objects.

Cloning an Object

To see how serialization works, I'll implement a Clone method in the Home class. Cloning is the process of making an exact copy of an object. You can use serialization to build a Clone method with very little code.

All you need to do is serialize our object into a byte stream, and then deserialize it to create a new instance of the class—an exact copy of the original, since the deserialization process will create it by using the data from the original object.

Before working with serialization, it is always a good idea to import the appropriate namespaces. This helps keep your code readable as you work with stream and formatter objects. Add the following to the top of the code module containing the Home class:

Imports System.IO

Imports System.Runtime.Serialization.Formatters.Binary

The following code shows the implementation of the Clone method for the Home class:

  Public Function Clone() As Home

    Dim m As New MemoryStream()

    Dim b As New BinaryFormatter()

    b.Serialize(m, Me)

    m.Position = 0

    Return b.Deserialize(m)

  End Function

This method is a function that returns a new instance of the Home class that is an exact copy of the current object. To make this copy, start by declaring both a new MemoryStream and BinaryFormatter object. The MemoryStream object is simply a data stream that resides entirely in memory. The BinaryFormatter object will handle the serialization and deserialization process.

You can then call the Serialize method on the BinaryFormatter to serialize the state data into the MemoryStream—converting all our data into a single stream of bytes in memory.

    b.Serialize(m, Me)

The MemoryStream object has the concept of a current position or cursor within the stream of bytes. As you write data into the stream, that position is always updated to be at the end of the stream. Once you're done writing the object's data into the stream, you need to reset the position to the beginning of the stream so your code can read the data back out. To do this, set the Position property to 0.

Finally you can create a new instance of the Home class—populated with the serialized data—by calling the Deserialize method on the BinaryFormatter object:

    Return b.Deserialize(m)

This is the object that is returned as the result of the function—an exact copy of the original object.

Serializing an Object Graph

As I mentioned earlier, deep serialization not only converts an object's data into a byte stream, but it also includes the data for any objects referenced by the object being serialized. This means that the process may result in serializing and deserializing many objects into and out of the byte stream.

To see how this works, add a new class to the project:

<Serializable()> _

Public Class Room

  Private mstrName As String

  Private mintSize As Integer

  Public Sub New(ByVal Name As String, ByVal Size As Integer)

    mstrName = Name

    mintSize = Size

  End Sub

  Public ReadOnly Property Name() As String

    Get

      Return mstrName

    End Get

  End Property

  Public ReadOnly Property Size() As Integer

    Get

      Return mintSize

    End Get

  End Property

End Class

This class represents a room within a house. Notice that it is marked with the <Serializable ()> attribute as well. In order to serialize an entire graph of objects, all the classes within the graph must be marked with this attribute—otherwise a runtime error will occur when the BinaryFormatter attempts to serialize the child object.

The Home class can then be enhanced to contain a collection of Room objects. The Hashtable class from System.Collections is serializable, so it is a good candidate:

<Serializable()> _

Public Class Home

  Private mstrAddress As String

  Private mintSize As Integer

  Private mdtBuilt As Date

  Private mcolRooms As New Hashtable()

  Public Function Rooms() As Hashtable

    Return mcolRooms

  End Function

In a real application, you should implement a custom collection to store the Room objects, but this code will work for a demonstration of serialization.

The Clone method you implemented earlier will now automatically handle the serialization process for the Room objects. No change is required. The BinaryFormatter object will automatically pick up on the mcolRooms variable and will serialize the Hashtable object and all the objects it contains.

To see this in action, add the following code behind the Load event of the application's form:

  Private Sub Form1_Load(ByVal sender As System.Object, _

      ByVal e As System.EventArgs) Handles MyBase.Load

    Dim objHome As New Home("123 Somestreet, Sometown", 1900, #1/1/1972#)

    Dim objRoom As Room

    objRoom = New Room("Kitchen", 100)

    objHome.Rooms.Add(objRoom.Name, objRoom)

    objRoom = New Room("Living", 150)

    objHome.Rooms.Add(objRoom.Name, objRoom)

    Dim objNewHome As Home = objHome.Clone

    MsgBox(objNewHome.Rooms.Count)

  End Sub

When this application is run, it will create a Home object and add two Room objects to its collection. The Home object is then cloned, creating a second Home object that is an exact copy of the first. When the message box is displayed, it will show that there are two Room objects in the collection; the two child objects were automatically cloned along with the Home object itself.

Preventing Serialization

There are times when you may not want a variable or object reference to be serialized. 

This is particularly valuable in 'pruning' the object graph to prevent all object references from being serialized. Sometimes object graphs can include interlinked references to hundreds or thousands of objects in memory—and when you go to serialize one object, you often want to prevent the automatic serialization of all those objects.

Also, your object may reference other objects that are not marked with the <Serializable()> attribute. Attempting to serialize such an object will result in a runtime error, and so it is important that the serializer skip over that object reference rather than attempt to serialize it.

You may also want to prevent serialization of specific variables that don't refer to an object. Perhaps a variable that contains an image or some other large data element is too expensive to be copied.

To prevent serialization of an instance variable, you can use the <NonSerialized()> attribute on the variable declaration. This will prevent the serializer from making any attempt to copy that variable into the byte stream—including preventing it from attempting to serialize a child object if the variable holds an object reference.

For example, to prevent the collection of Room objects from being serialized, you can add this attribute to the declaration of the collection variable:

  <NonSerialized()> Private mcolRooms As New Hashtable()

With this change, the BinaryFormatter will ignore this variable during the serialization process, meaning that neither the Hashtable nor any of the child Room objects will be serialized into the byte stream.

When you do this, you need to keep in mind that there are side effects. In particular, when the byte stream is deserialized to create a new Home object, the Hashtable will not be recreated (which we would expect), but the side effect is that the new Home object's initialization code is not run—meaning that the new Home object will contain no Hashtable object at all. The mcolRooms variable will hold a null reference.

This means that to prevent accidental runtime errors when the Rooms method is accessed, you need to check to see if the mcolRooms variable is Nothing:

  Public Function Rooms() As Hashtable

    If IsNothing(mcolRooms) Then mcolRooms = New Hashtable()

    Return mcolRooms

  End Function

If this code is not added, the function will return Nothing, which will most likely cause an error in the calling code.

Now when the application is run, the message box will display a 0, indicating that the copy of the Home object has no child Room objects. The Home object itself was copied via serialization, but the entire collection of child objects has been 'pruned' by use of the <NonSerialized()> attribute.

Implementing ISerializable

The automatic serialization support built into .NET is very powerful and very easy. However, there are times when you may find the automatic behavior of serialization to be too limiting. In such a case, you can opt for the more advanced approach by implementing the ISerializable interface.

Implementing ISerializable allows you to control the serialization and deserialization process to a much greater degree, but at the cost of having to write quite a bit more code within your classes.

You can think of serialization within Microsoft® Visual Basic® .NET as being somewhat similar to the Persistable property setting on a class in Microsoft® Visual Basic® 6.0. In Visual Basic 6.0 you could mark a class as Persistable and then implement Class_ReadProperties and Class_WriteProperties methods to copy the data within your object into and out of a PropertyBag object supplied by the Visual Basic runtime. The runtime would then use this PropertyBag to store your object's data in a file, and send it in an MSMQ message or whatever was required.

When you choose to implement ISerializable instead of using the <Serializable()> attribute, you are basically choosing to write the same kind of code to populate and read a property bag object as you would have in Visual Basic 6.0.

To write your own serialization in the Home class, you need to implement the ISerializable interface from the System.Runtime.Serialization namespace:

Imports System.IO

Imports System.Runtime.Serialization

Imports System.Runtime.Serialization.Formatters.Binary

<Serializable()> _

Public Class Home

  Implements ISerializable

This interface only defines a single method, GetObjectData. However, you also need to implement a specific constructor method for your class as well or deserialization will fail.

The GetObjectData method is automatically called when your object is serialized. It is this method that must populate a property bag with all the data from your object:

  Public Sub GetObjectData(ByVal Info As SerializationInfo, _

      ByVal Context As StreamingContext) _

      Implements ISerializable.GetObjectData

    With Info

      .AddValue("Address", mstrAddress)

      .AddValue("Size", mintSize)

      .AddValue("Built", mdtBuilt)

    End With

  End Sub

The Info parameter is of type SerializationInfo and it acts very much like a property bag object in that it stores a set of name-value pairs. You can put any values into this property bag that you'd like, which provides a great deal of flexibility in terms of what is and is not serialized as part of your object.

When the object is deserialized, a special constructor method will be invoked. This constructor method must be supplied or deserialization will fail with a runtime error. Typically, you won't want this constructor visible to normal users of your class—and that is fine since the constructor can be declared as Private, and the serialization mechanism will still be able to call it:

  Private Sub New(ByVal Info As SerializationInfo, _

      ByVal Context As StreamingContext)

    With Info

      mstrAddress = .GetString("Address")

      mintSize = .GetInt32("Size")

      mdtBuilt = .GetDateTime("Built")

      mcolRooms = New Hashtable()

    End With

  End Sub

Notice that this code also includes the creation of a new Hashtable object for the mcolRooms variable, thus eliminating the need for the extra check in the Rooms method:

  Public Function Rooms() As Hashtable

    Return mcolRooms

  End Function

This illustrates how implementing ISerializable can increase the level of control you have over the process, but also shows how much extra code may be involved as compared to using the default behavior.

Controlling the Deserialization Class

Not only can you control the data that is placed into the property bag during the serialization process, but you can even control the class of object that is created during deserialization. This means that you can serialize an object of one type, and get an entirely different type of object as a result of deserialization.

Figure 4. Deserializing into a different type
This can be particularly useful when passing objects across the network. You may start with an object on the server that contains data access code, but serializes itself across the network to the client into a different type of object that contains business logic and validation code.

As an example, add a HomeInfo class to the project:

Imports System.Runtime.Serialization

<Serializable()> _

Public Class HomeInfo

  Implements ISerializable

  Private mstrAddress As String

  Private mintSize As Integer

  Private mintAge As Integer

  Private Sub New(ByVal Info As SerializationInfo, _

      ByVal Context As StreamingContext)

    Debug.WriteLine("here")

    With Info

      mstrAddress = .GetString("Address")

      mintSize = .GetInt32("Size")

      mintAge = .GetInt32("Age")

    End With

  End Sub

  Public Sub GetObjectData(ByVal Info As SerializationInfo, _

      ByVal Context As StreamingContext) _

      Implements ISerializable.GetObjectData

    With Info

      .AddValue("Address", mstrAddress)

      .AddValue("Size", mintSize)

      .AddValue("Age", mintAge)

    End With

  End Sub

  Public ReadOnly Property Address() As String

    Get

      Return mstrAddress

    End Get

  End Property

  Public ReadOnly Property Size() As Integer

    Get

      Return mintSize

    End Get

  End Property

  Public ReadOnly Property Age() As Integer

    Get

      Return mintAge

    End Get

  End Property

End Class

This class is similar to the Home class, but contains no collection of Room objects and does not keep track of the date the home was built. Instead, it merely contains the derived Age value. It is a simplified version of the Home object.

You can make the Home object deserialize into a HomeInfo object, rather than a Home object, by changing the GetObjectData method in the Home class:

  Public Sub GetObjectData(ByVal Info As SerializationInfo, _

      ByVal Context As StreamingContext) _

      Implements ISerializable.GetObjectData

    With Info

      .FullTypeName = "vbSerialization.HomeInfo"

      .AddValue("Address", mstrAddress)

      .AddValue("Size", mintSize)

      .AddValue("Age", Age)

    End With

  End Sub

By setting the FullTypeName property of the Info object, you indicate that the deserialization process should create an object of a type other than Home. In this case the deserialization process will create and populate an object of type HomeInfo. Notice that the full type name is used—including the project's namespace—and this is a requirement of the serialization mechanism.

Also notice that the mdtBuilt variable is no longer inserted into the property bag, but instead the Age value is inserted. This is critical, since the constructor in the HomeInfo class expects this Age value to be supplied. If it is not supplied, a runtime error will occur as the object is deserialized.

Now that deserialization creates an object of type HomeInfo, the Clone method needs to be updated:

  Public Function Clone() As HomeInfo

    Dim m As New MemoryStream()

    Dim b As New BinaryFormatter()

    b.Serialize(m, Me)

    m.Position = 0

    Return b.Deserialize(m)

  End Function

The serialization and deserialization process doesn't change, but the resulting data type of the deserialization process is now different, and so the function needs to return an object of type HomeInfo.

To test this code, make the following changes to the code in the Load event of the form:

  Private Sub Form1_Load(ByVal sender As System.Object, _

      ByVal e As System.EventArgs) Handles MyBase.Load

    Dim objHome As New Home("123 Somestreet, Sometown", 1900, #1/1/1972#)

    Dim objRoom As Room

    objRoom = New Room("Kitchen", 100)

    objHome.Rooms.Add(objRoom.Name, objRoom)

    objRoom = New Room("Living", 150)

    objHome.Rooms.Add(objRoom.Name, objRoom)

    Dim objNewHome As HomeInfo = objHome.Clone

    MsgBox(objNewHome.Age)

  End Sub

Since the result of the Clone method is now a HomeInfo object, the declaration of the objNewHome variable needs to be changed. Additionally, the message box will now display the Age property to illustrate that the Age value has been successfully transferred from the original Home object to the new HomeInfo object.

Conclusion

The object serialization capabilities built into the .NET Framework and available to Visual Basic .NET developers are very powerful and very simple to use. This easy ability to serialize objects into any type of stream object is powerful, since it allows us to efficiently store objects in memory, in files on disk, or pass them across the network using TCP/IP sockets or .NET Remoting.

